

TUTORIEL : CREER UNE FICHE D'ÉVALUATION AVEC CALCUL AUTOMATIQUE DE LA NOTE ET REFERENCE A UN BAREME.

Nous allons créer une fiche d'évaluation BAC LGT en Pentabond avec une recherche d'automatisation des calculs pour déterminer la note finale sur 20.

✚ Etape 1 : Reproduire les 3 tableaux suivants.

-Notre fiche s'organise en 2 feuilles (ou onglet) : 1 feuille « **Fiche Protocole** » dans laquelle nous calculerons la note des élèves à partir de nos critères d'évaluation, et 1 fiche « **Barèmes** » où l'on mettra le tableau source qui contiendra le barème servant de référence à la note de performance.

1-1 Reproduire le 1^{er} Tableau sur la « **Feuil1** » : « **Fiche Protocole** »

FEUILLE CALCUL NOTE BAC LGT PENTABOND																				
Note	Nom	Prenom	sexe	classe	PROJET	PERFORMANCES /10 pts							CUMUL : moyenne des 3 meilleurs essais /4 pts		ECART AU PROJET /2 pts		MAITRISE : Régularité des bonds et coordination /4 pts			
						Essai 1	Essai 2	Essai 3	Essai 4	Essai 5	Essai 6	Perf. Max	Note perf /10	Moy. L1-L2-L3	Note /4	Ecart au projet		Note /2		

1-2 Reproduire le 2^{ème} Tableau sur la « **Feuil2** » : « **Baremes** »

Pentabond Perf				Pentabond Moyenne			
FILLES		GARÇONS		FILLES		GARÇONS	
PERF	NOTE	PERF	NOTE	PERF	NOTE	PERF	NOTE
0,00	0	0,00	0	0,00	0	0,00	0
6,87	0,5	8,00	0,5	6,30	0,2	7,50	0,2
6,93	1	8,25	1	6,50	0,4	7,75	0,4
7,00	1	8,50	1	6,87	0,6	8,00	0,6
7,21	1,5	8,75	1,5	7,00	0,8	8,50	0,8
7,42	1,5	9,05	1,5	7,42	1	9,05	1
7,63	2	9,30	2	7,83	1,2	9,55	1,2
7,83	2	9,55	2	8,22	1,4	10,07	1,4
8,01	2,5	9,80	2,5	8,62	1,6	10,57	1,6
8,22	2,5	10,07	2,5	9,00	1,8	11,06	1,8
8,43	3	10,30	3	9,39	2	11,56	2
8,62	3	10,57	3	9,76	2,2	12,05	2,2
8,84	3,5	11,80	3,5	10,14	2,4	12,53	2,4
9,00	3,5	11,06	3,5	10,49	2,6	13,00	2,6
9,20	4	11,32	4	10,86	2,8	13,47	2,8
9,39	4	11,56	4	11,20	3	13,93	3

Etape 2 : Sélectionner automatiquement la meilleure perf.

UTILISATION DES FONCTIONS :

Fonction GRANDE.VALEUR pour déterminer la ou les plus grandes valeurs d'une liste (plage)

Syntaxe

GRANDE.VALEUR(matrice, k)

La syntaxe de la fonction GRANDE.VALEUR contient les argument suivants :

- 1) matrice** Obligatoire. Représente la matrice ou la plage de données dans laquelle vous recherchez la k-ième plus grande valeur.
- 2) k** Obligatoire. Représente, dans la matrice ou la plage de cellules, la position de la valeur à renvoyer, déterminée à partir de la valeur la plus grande.

SI

Description

La fonction **SI** renvoie une valeur si la condition que vous spécifiez est VRAI, et une autre valeur si cette condition est FAUX.

Par exemple, la formule **=SI(A1>10;"Plus de 10";"10 ou moins")** renvoie "Plus de 10" si A1 est supérieur à 10, et "10 ou moins" si A1 est inférieur ou égal à 10.

Syntaxe

SI(test_logique; [valeur_si_vrai]; [valeur_si_faux])

ESTERREUR

Syntaxe

ESTERREUR (valeur)

Valeur à tester. La valeur peut être une cellule vide, une erreur, une expression logique, un texte, un nombre, une valeur de référence ou un nom faisant référence à l'un de ces éléments.


Exemple

La fonction **ESTERREUR** est utile dans les formules pour tester le résultat d'un calcul. Lorsqu'elle est associée à la fonction **SI**, elle permet de repérer des erreurs dans les formules (voir les exemples suivants).

2-1 FONCTION GRANDE.VALEUR POUR TROUVER LA PLUS HAUTE VALEUR DE PERF

Pour sélectionner le meilleur essai sans élan, on va utiliser la fonction **Grande.Valeur** (fonction qui permet de déterminer la valeur la plus élevée dans une série de valeurs).

Pour se faire, en cellule M5 de la « **fiche protocole** » on tape dans la barre de formule « =grande.valeur » et on remplit les paramètres, ou alors on va dans Fx et on recherche la fonction Grande.Valeur :



Il faut maintenant indiquer la matrice, c'est-à-dire la plage de cellule ou « Excel » doit rechercher cette plus grande valeur.

Et on va sélectionner les cellules qui correspondent aux essais sans élan du 1^{er} élève à savoir les cellules **G5 à L5**

On tape maintenant « ; » et on précise le « **k** » de la formule c'est-à-dire la 1^{ère} plus grande valeur et on tape « **1** » (il serait possible de choisir la 2^{ème}, 3^{ème}, voir n^{ème} plus grande valeur). On valide notre formule en tapant sur la touche « **Entrer** ».

La formule visible dans la barre de formule est donc la suivante :

=GRANDE.VALEUR(G5:L5;1)

L'erreur « #NOMBRE! » apparaît, car aucune cellule n'étant renseignée, la fonction ne peut trouver de valeur. Elle retourne donc une erreur au lieu de la valeur trouvée

Nous allons donc traiter cette erreur grâce au cumul des fonctions **SI et ESTERREUR**

FONCTION SI

SI(test_logique; [valeur_si_vrai]; [valeur_si_faux])

On veut tester : **Si (il y a une erreur ; alors on n'affiche rien ; sinon on affiche la valeur trouvée)**

FONCTION ESTERREUR

ESTERREUR (valeur à tester), elle retourne **VRAI** si une erreur apparaît ou **FAUX** s'il n'y a pas d'erreur

Ici on teste la valeur retournée par la fonction **GRANDE.VALEUR(G5:L5;1)**

Donc : ESTERREUR (GRANDE.VALEUR(G5:L5;1)), qui sera notre **test_logique**

Ce qui nous donne

SI(ESTERREUR (GRANDE.VALEUR(G5:L5;1));"" ;GRANDE.VALEUR(G5:L5;1))

On n'oublie pas d'appliquer la formule au reste de la colonne en faisant un « Copier » / « Coller Formule »

2-2 FONCTION GRANDE.VALEUR POUR TROUVER LA MOYENNE DES 3 PLUS HAUTES VALEURS DE PERF

De la même manière que précédemment, nous appliquerons en cellule O5 (cellule où doit apparaître la moyenne des 3 meilleures perf), la fonction Grande.Valeur, mais en modifiant 1 paramètre :

GRANDE.VALEUR(G5:L5;**{1;2;3}**). Nous précisons, entre accolades, que nous voulons récupérer la 1^{ère}, 2^{ème} et 3^{ème} valeurs les plus hautes de la série.

Puis nous calculerons la moyenne de ces 3 valeurs par la fonction MOYENNE :

■ MOYENNE

Description

Renvoie la moyenne (arithmétique) des arguments. Par exemple, si la plage A1:A20 contient des nombres, la formule =MOYENNE(A1:A20) renvoie la moyenne de ces nombres.

Ou = MOYENNE(A1 ;A2 ;A3 ;...)

Syntaxe

MOYENNE(nombre1; [nombre2]; ...)

La syntaxe de la fonction MOYENNE contient les argument suivants :

nombre1 Obligatoire. Premier nombre, référence de cellule ou plage dont vous voulez obtenir la moyenne.

nombre2; ... Facultatif. Nombres, références de cellules ou plages supplémentaires dont vous voulez obtenir la moyenne (255 maximum).

=MOYENNE(GRANDE.VALEUR(G5:L5;**{1;2;3}**))

Comme tout à l'heure, nous ferons un contrôle d'erreur, afin de ne pas en afficher dans les cellules Excel, grâce aux fonctions SI et ESTERREUR.

Ce qui nous donnera :

=SI(ESTERREUR(MOYENNE(GRANDE.VALEUR(G6:L6;**{1;2;3}**)));"";MOYENNE(GRANDE.VALEUR(G6:L6;**{1;2;3}**)))

On n'oublie pas d'appliquer la formule au reste de la colonne en faisant un « Copier » / « Coller Formule »

Etape 3 : Utilisation des fonctions SI et RECHERCHE pour attribuer la note de Performance.

Voici la partie la plus intéressante et utile de ce tutoriel, faire en sorte qu'Excel puisse calculer automatiquement la note de performance de l'élève en fonction de son sexe (référence à un barème féminin ou masculin).

Pour ce faire nous allons combiner dans la même formule, la fonction **SI** (qui permettra de choisir le bon barème en fonction du sexe) et la fonction **RECHERCHE** (qui permettra de comparer la performance de l'élève au barème et de lui attribuer la note correspondante).

RECHERCHE (dans notre cas)

Description

Elle est capable de rechercher une valeur exacte ou approximative (inférieure) et de renvoyer la valeur de la colonne suivante

Syntaxe

RECHERCHE(Valeur cherchée; Matrice de recherche (avec 1ere colonne pour comparer valeurs et deuxième colonne est la valeur cible renvoyée))

Exemple

On veut chercher la valeur de la cellule M5 par exemple et ceci dans la colonne A si c'est une Fille et C si c'est un garçon.

On veut, une fois trouver, renvoyer la valeur (Note) qui correspond et qui se trouve dans la colonne suivante

Pour un Garçon :

RECHERCHE(M5;Baremes!\$C\$6:\$D\$47) : on recherche dans les cellules de C6 à C47 la valeur de la cellule M5, et on renvoie la valeur trouvée dans la colonne suivante de D6 à D47

Pour une Fille :

RECHERCHE(M5;Baremes!\$A\$6:\$B\$47) : on recherche dans les cellules de A6 à A47 la valeur de la cellule M5, et on renvoie la valeur trouvée dans la colonne suivante de B6 à B47

Pentabond Perf				
	FILLES		GARCONS	
	PERF	NOTE	PERF	NOTE
6	0,00	0	0,00	0
7	6,87	0,5	8,00	0,5
8	6,93	1	8,25	1
9	7,00	1	8,50	1
10	7,21	1,5	8,75	1,5
11	7,42	1,5	9,05	1,5
12	7,63	2	9,30	2
13	7,83	2	9,55	2
14	8,01	2,5	9,80	2,5
15	8,22	2,5	10,07	2,5
16	8,43	3	10,30	3
17	8,62	3	10,57	3
18	8,84	3,5	11,80	3,5
19	9,00	3,5	11,06	3,5
20	9,20	4	11,32	4
21	9,39	4	11,56	4

Attention

La fonction **RECHERCHE** ne marche que si les valeurs de votre tableau source (tableau où il y a le barème) sont classées par ordre croissant si ce n'est pas le cas vous obtiendrez un message d'erreur !!!

Nous venons de voir qu'il y a deux possibilités pour trouver la valeur : l'une si c'est une fille et l'autre si c'est un garçon, car il y a deux barèmes différents

Nous allons donc cumuler la fonction RECHERCHE avec la fonction SI :

SI(c'est une fille ; alors note est... ; sinon note est...)

Pour se faire en cellule N5 de la « **fiche protocole** » (colonne « **note de performance** ») on rentre la formule suivante :

=SI(D5="F";RECHERCHE(M5;Baremes!A6:B47);RECHERCHE(M5;Baremes!C6:D47))

SI en cellule **D5** (cellule déterminant le sexe de notre élève) on la lettre « F » alors on va rechercher la performance sur la feuille « Barèmes ! », et dans la plage A6 à B47 (barème féminin).

Sinon, on va chercher dans la plage de C6 à D47 (barème masculin)

Deux problèmes :

- 1- Lorsque nous allons copier la formule sur les autres lignes, toutes les valeurs de cellules vont être **incrémentées**.

On va donc bloquer celles du tableau de recherche car elles ne changent jamais.

Pour fixer la plage lors de la copie on ajoute le signe « \$ »

=SI(D5="F";RECHERCHE(M5;Baremes!\$A\$6:\$B\$47);RECHERCHE(M5;Baremes!\$C\$6:\$D\$47))

- 2- Une **erreur de recherche apparait tant que rien n'est rempli (cellule M5 vide)**.

On va donc contrôler l'erreur en utilisant une autre technique.

On va tester avec SI : si M5 n'est pas vide, on applique la formule, sinon on ne met rien

=SI(NB(M5)>0;SI(D5="F";RECHERCHE(M5;Baremes!\$A\$6:\$B\$47);RECHERCHE(M5;Baremes!\$C\$6:\$D\$47));'')

NOUS AVONS UNE FORMULE COMPLETE IMBRIQUEE.

On n'oublie pas d'appliquer la formule au reste de la colonne en faisant un « **Copier** » / « **Coller Formule** »

Vous pouvez maintenant essayer votre fiche d'évaluation en rentrant les valeurs des essais et en faisant varier le sexe de votre élève pour vérifier que les 2 barèmes sont bien pris en compte (masculin et féminin).

POUR ALLER PLUS LOIN

Cette démarche est à refaire pour le calcul automatique des notes de la moyenne des 3 meilleures perf !!!

Donc nouveau tableau de notes à faire, et nouvelles formules !!!

Nous pouvons faire un copier coller en P5 de N5, et modifier quelques valeurs !

=SI(NB(O5)>0;SI(D5="F";RECHERCHE(O5;Baremes!\$F\$6:\$G\$27);RECHERCHE(O5;Baremes!\$H\$6:\$I\$27));"")

Etape 4 : Utilisation de la fonction SOMME.

Calcul de la « **Note sur 20** », en utilisant la fonction SOMME.

La note sur 20 de l'élève résulte de l'addition des colonnes N, P, R et S qui correspondent à :

- La note de la meilleure perf
- La note de la moyenne des 3 meilleures perf
- La note de l'écart au projet
- La note de maîtrise

Pour se faire en cellule A5 de la « **fiche évaluation** » on rentre la formule suivante :

=SOMME(N5;P5;R5;S5)

Et bien sur, pour éviter de voir des 0 s'afficher, nous faisons un contrôle :

=SI(SOMME(N5;P5;R5;S5)=0;"";SOMME(N5;P5;R5;S5))

On n'oublie pas d'appliquer la formule au reste de la colonne en faisant un « **Copier** » / « **Coller Formule** »

C'est ainsi que s'achève ce tutoriel de niveau 2. Pour ceux qui souhaiteraient apprendre davantage de chose sur la mise en forme sous Excel, je vous renvoie vers le tutoriel de niveau 1.

Quoi qu'il en soit, rappelez-vous que vous devez construire un outil qui a du sens pour vous et qu'il ne s'agit là que d'un exemple.